

# NAG Fortran Library Routine Document

## F04CBF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

F04CBF computes the solution to a complex system of linear equations  $AX = B$ , where  $A$  is an  $n$  by  $n$  band matrix, with  $k_l$  sub-diagonals and  $k_u$  super-diagonals, and  $X$  and  $B$  are  $n$  by  $r$  matrices. An estimate of the condition number of  $A$  and an error bound for the computed solution are also returned.

### 2 Specification

```

SUBROUTINE F04CBF (N, KL, KU, NRHS, AB, LDAB, IPIV, B, LDB, RCOND,
1                ERRBND, IFAIL)
    INTEGER          N, KL, KU, NRHS, LDAB, IPIV(*), LDB, IFAIL
    double precision RCOND, ERRBND
    complex*16       AB(LDAB,*), B(LDB,*)

```

### 3 Description

The  $LU$  decomposition with partial pivoting and row interchanges is used to factor  $A$  as  $A = PLU$ , where  $P$  is a permutation matrix,  $L$  is the product of permutation matrices and unit lower triangular matrices with  $k_l$  sub-diagonals, and  $U$  is upper triangular with  $(k_l + k_u)$  super-diagonals. The factored form of  $A$  is then used to solve the system of equations  $AX = B$ .

### 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: <http://www.netlib.org/lapack/lug>

Higham N J (2002) *Accuracy and Stability of Numerical Algorithms* (2nd Edition) SIAM, Philadelphia

### 5 Parameters

- |    |   |              |
|----|---|--------------|
| 1: | N – INTEGER   | <i>Input</i> |
|    | <i>On entry:</i> the number of linear equations $n$ , i.e., the order of the matrix $A$ . |              |
|    | <i>Constraint:</i> $N \geq 0$ .   |              |
| 2: | KL – INTEGER  | <i>Input</i> |
|    | <i>On entry:</i> the number of sub-diagonals $k_l$ , within the band of $A$ .             |              |
|    | <i>Constraint:</i> $KL \geq 0$ .  |              |
| 3: | KU – INTEGER  | <i>Input</i> |
|    | <i>On entry:</i> the number of super-diagonals $k_u$ , within the band of $A$ .           |              |
|    | <i>Constraint:</i> $KU \geq 0$ .  |              |

- 4: NRHS – INTEGER *Input*  
*On entry:* the number of right-hand sides  $r$ , i.e., the number of columns of the matrix  $B$ .  
*Constraint:* NRHS  $\geq 0$ .
- 5: AB(LDAB,\*) – **complex\*16** array *Input/Output*  
**Note:** the second dimension of the array AB must be at least  $\max(1, N)$ .  
*On entry:* the  $n$  by  $n$  coefficient matrix  $A$  in band storage, in rows  $KL + 1$  to  $2 \times KL + KU + 1$ ; rows 1 to  $KL$  of the array need not be set. The  $j$ th column of  $A$  is stored in the  $j$ th column of the array AB as follows:  

$$AB(k_l + k_u + 1 + i - j, j) = a_{ij} \quad \text{for } \max(1, j - k_u) \leq i \leq \min(n, j + k_l)$$
See Section 8 below for further details.  
*On exit:* if IFAIL  $\geq 0$ , details of the factorization  $A = PLU$ ;  $U$  is stored as an upper triangular band matrix with  $KL + KU$  super-diagonals in rows 1 to  $KL + KU + 1$ , and the multipliers used during the factorization are stored in rows  $KL + KU + 2$  to  $2 \times KL + KU + 1$ .
- 6: LDAB – INTEGER *Input*  
*On entry:* the first dimension of the array AB as declared in the (sub)program from which F04CBF is called.  
*Constraint:* LDAB  $\geq 2 \times KL + KU + 1$ .
- 7: IPIV(\*) – INTEGER array *Output*  
**Note:** the dimension of the array IPIV must be at least  $\max(1, N)$ .  
*On exit:* if IFAIL  $\geq 0$ , the pivot indices that define the permutation matrix  $P$ ; at the  $i$ th step row  $i$  of the matrix was interchanged with row IPIV( $i$ ). IPIV( $i$ ) =  $i$  indicates a row interchange was not required.
- 8: B(LDB,\*) – **complex\*16** array *Input/Output*  
**Note:** the second dimension of the array B must be at least  $\max(1, \text{NRHS})$ . To solve the equations  $Ax = b$ , where  $b$  is a single right-hand side, B may be supplied as a one-dimensional array with length LDB =  $\max(1, N)$ .  
*On entry:* the  $n$  by  $r$  matrix of right-hand sides  $B$ .  
*On exit:* if IFAIL = 0 or  $N + 1$ , the  $n$  by  $r$  solution matrix  $X$ .
- 9: LDB – INTEGER *Input*  
*On entry:* the first dimension of the array B as declared in the (sub)program from which F04CBF is called.  
*Constraint:* LDB  $\geq \max(1, N)$ .
- 10: RCOND – **double precision** *Output*  
*On exit:* if IFAIL  $\geq 0$ , an estimate of the reciprocal of the condition number of the matrix  $A$ , computed as  $RCOND = \left( \|A\|_1 \|A^{-1}\|_1 \right)^{-1}$ .
- 11: ERRBND – **double precision** *Output*  
*On exit:* if IFAIL = 0 or  $N + 1$ , an estimate of the forward error bound for a computed solution  $\hat{x}$ , such that  $\|\hat{x} - x\|_1 / \|x\|_1 \leq \text{ERRBND}$ , where  $\hat{x}$  is a column of the computed solution returned in the array B and  $x$  is the corresponding column of the exact solution  $X$ . If RCOND is less than **machine precision**, then ERRBND is returned as unity.

## 12: IFAIL – INTEGER

Input/Output

*On entry:* IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

*On exit:* IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL < 0 and IFAIL  $\neq$  -999

If IFAIL = - $i$ , the  $i$ th argument had an illegal value.

IFAIL = -999

Allocation of memory failed. The **double precision** allocatable memory required is  $N$ , and the **complex\*16** allocatable memory required is  $2 \times N$ . In this case the factorization and the solution  $X$  have been computed, but RCOND and ERBND have not been computed.

IFAIL > 0 and IFAIL  $\leq N$

If IFAIL =  $i$ ,  $u_{ii}$  is exactly zero. The factorization has been completed, but the factor  $U$  is exactly singular, so the solution could not be computed.

IFAIL =  $N + 1$

RCOND is less than **machine precision**, so that the matrix  $A$  is numerically singular. A solution to the equations  $AX = B$  has nevertheless been computed.

## 7 Accuracy

The computed solution for a single right-hand side,  $\hat{x}$ , satisfies an equation of the form

$$(A + E)\hat{x} = b,$$

where

$$\|E\|_1 = O(\epsilon)\|A\|_1$$

and  $\epsilon$  is the **machine precision**. An approximate error bound for the computed solution is given by

$$\frac{\|\hat{x} - x\|_1}{\|x\|_1} \leq \kappa(A) \frac{\|E\|_1}{\|A\|_1},$$

where  $\kappa(A) = \|A^{-1}\|_1 \|A\|_1$ , the condition number of  $A$  with respect to the solution of the linear equations. F04CBF uses the approximation  $\|E\|_1 = \epsilon \|A\|_1$  to estimate ERBND. See Section 4.4 of Anderson *et al.* (1999) for further details.

## 8 Further Comments

The band storage scheme for the array AB is illustrated by the following example, when  $n = 6$ ,  $k_l = 1$ , and  $k_u = 2$ . Storage of the band matrix  $A$  in the array AB:

*	*	*	+	+	+
*	*	$a_{13}$	$a_{24}$	$a_{35}$	$a_{46}$
*	$a_{12}$	$a_{23}$	$a_{34}$	$a_{45}$	$a_{56}$
$a_{11}$	$a_{22}$	$a_{33}$	$a_{44}$	$a_{55}$	$a_{66}$
$a_{21}$	$a_{32}$	$a_{43}$	$a_{54}$	$a_{65}$	*

Array elements marked \* need not be set and are not referenced by the routine. Array elements marked + need not be set, but are defined on exit from the routine and contain the elements  $u_{14}$ ,  $u_{25}$  and  $u_{36}$ .

The total number of floating-point operations required to solve the equations  $AX = B$  depends upon the pivoting required, but if  $n \gg k_l + k_u$  then it is approximately bounded by  $O(nk_l(k_l + k_u))$  for the factorization and  $O(n(2k_l + k_u)r)$  for the solution following the factorization. The condition number estimation typically requires between four and five solves and never more than eleven solves, following the factorization.

In practice the condition number estimator is very reliable, but it can underestimate the true condition number; see Section 15.3 of Higham (2002) for further details.

The real analogue of F04CBF is F04BBF.

## 9 Example

To solve the equations

$$AX = B,$$

where  $A$  is the band matrix

$$A = \begin{pmatrix} -1.65 + 2.26i & -2.05 - 0.85i & 0.97 - 2.84i & & 0 \\ 0.00 + 6.30i & -1.48 - 1.75i & -3.99 + 4.01i & 0.59 - 0.48i & \\ & 0 & -0.77 + 2.83i & -1.06 + 1.94i & 3.33 - 1.04i \\ & 0 & & 0 & 4.48 - 1.09i \\ & & & & -0.46 - 1.72i \end{pmatrix}$$

and

$$B = \begin{pmatrix} -1.06 + 21.50i & 12.85 + 2.84i \\ -22.72 - 53.90i & -70.22 + 21.57i \\ 28.24 - 38.60i & -20.73 - 1.23i \\ -34.56 + 16.73i & 26.01 + 31.97i \end{pmatrix}.$$

An estimate of the condition number of  $A$  and an approximate error bound for the computed solutions are also printed.

### 9.1 Program Text

**Note:** the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F04CBF Example Program Text
*      Mark 21 Release. NAG Copyright 2004.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, KLMAX, KUMAX, NRHSMX
      PARAMETER        (NMAX=8, KLMAX=4, KUMAX=4, NRHSMX=2)
      INTEGER          LDAB, LDB
      PARAMETER        (LDAB=2*KLMAX+KUMAX+1, LDB=NMAX)
*      .. Local Scalars ..
      DOUBLE PRECISION ERRBND, RCOND
      INTEGER          I, IERR, IFAIL, J, K, KL, KU, N, NRHS
*      .. Local Arrays ..
      COMPLEX *16      AB(LDAB,NMAX), B(LDB,NRHSMX)
      INTEGER          IPIV(NMAX)
      CHARACTER        CLABS(1), RLABS(1)
```

```

*      .. External Subroutines ..
EXTERNAL          F04CBF, X04DBF, X04DFF
*      .. Intrinsic Functions ..
INTRINSIC          MAX, MIN
*      .. Executable Statements ..
WRITE (NOUT,*) 'F04CBF Example Program Results'
WRITE (NOUT,*)
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) N, KL, KU, NRHS
IF (N.LE.NMAX .AND. KL.LE.KLMAX .AND. KU.LE.KUMAX .AND. NRHS.LE.
+   NRHSMX) THEN
*
*      Read A and B from data file
*
      K = KL + KU + 1
      READ (NIN,*) ((AB(K+I-J,J),J=MAX(I-KL,1),MIN(I+KU,N)),I=1,N)
      READ (NIN,*) ((B(I,J),J=1,NRHS),I=1,N)
*
*      Solve the equations AX = B for X
*
      IFAIL = -1
      CALL F04CBF(N,KL,KU,NRHS,AB,LDAB,IPIV,B,LDB,RCOND,ERRBND,IFAIL)
*
      IF (IFAIL.EQ.0) THEN
*
*      Print solution, estimate of condition number and approximate
*      error bound
*
          IERR = 0
          CALL X04DBF('General',' ',N,NRHS,B,LDB,'Bracketed','F7.4',
+                   'Solution','Integer',RLABS,'Integer',CLABS,80,0,
+                   IERR)
*
          WRITE (NOUT,*)
          WRITE (NOUT,*) 'Estimate of condition number'
          WRITE (NOUT,99999) 1.0D0/RCOND
          WRITE (NOUT,*)
          WRITE (NOUT,*)
+         'Estimate of error bound for computed solutions'
          WRITE (NOUT,99999) ERBND
      ELSE IF (IFAIL.EQ.N+1) THEN
*
*      Matrix A is numerically singular. Print estimate of
*      reciprocal of condition number and solution
*
          WRITE (NOUT,*)
          WRITE (NOUT,*) 'Estimate of reciprocal of condition number'
          WRITE (NOUT,99999) RCOND
*
          WRITE (NOUT,*)
          IERR = 0
          CALL X04DBF('General',' ',N,NRHS,B,LDB,'Bracketed','F7.4',
+                   'Solution','Integer',RLABS,'Integer',CLABS,80,0,
+                   IERR)
*
      ELSE IF (IFAIL.GT.0 .AND. IFAIL.LE.N) THEN
*
*      The upper triangular matrix U is exactly singular. Print
*      details of factorization
*
          WRITE (NOUT,*)
          IERR = 0
          CALL X04DFF(N,N,KL,KL+KU,AB,LDAB,'Bracketed','F7.4',
+                   'Details of factorization','None',RLABS,'None',
+                   CLABS,80,0,IERR)
*
*      Print pivot indices
*
          WRITE (NOUT,*)
          WRITE (NOUT,*) 'Pivot indices'

```

```

      WRITE (NOUT,99998) (IPIV(I),I=1,N)
    END IF
  ELSE
    WRITE (NOUT,*)
+    'One or more of NMAX, KLMAX, KUMAX or NRHSMX is too small'
  END IF
  STOP
*
99999 FORMAT (4X,1P,E9.1)
99998 FORMAT ((1X,7I11))
END

```

## 9.2 Program Data

F04CBF Example Program Data

```

      4              1              2              2      :N, KL, KU and NRHS

( -1.65,  2.26) ( -2.05, -0.85) (  0.97, -2.84)
(  0.00,  6.30) ( -1.48, -1.75) ( -3.99,  4.01) (  0.59, -0.48)
              ( -0.77,  2.83) ( -1.06,  1.94) (  3.33, -1.04)
              (  4.48, -1.09) ( -0.46, -1.72) :End matrix A

( -1.06, 21.50) ( 12.85,  2.84)
(-22.72,-53.90) (-70.22, 21.57)
( 28.24,-38.60) (-20.73, -1.23)
(-34.56, 16.73) ( 26.01, 31.97) :End matrix B

```

## 9.3 Program Results

F04CBF Example Program Results

Solution

```

              1              2
1  (-3.0000, 2.0000) ( 1.0000, 6.0000)
2  ( 1.0000,-7.0000) (-7.0000,-4.0000)
3  (-5.0000, 4.0000) ( 3.0000, 5.0000)
4  ( 6.0000,-8.0000) (-8.0000, 2.0000)

```

Estimate of condition number  
1.0E+02

Estimate of error bound for computed solutions  
1.2E-14

---